

URKATUA

Pili Lizaso
eta
Eustakio Arrojeria

URTEROKO ohiturari jarraituz, aurtan ere udako alean, hots, uztaila-abuztuko alean, oporrak datozela eta urtean zehar argitaratzen ditugun programak baino xamurrago den bat eskainiko dizuegu.

Xamurrago esaten dugunean ez gara programaren egiturari buruz ari; erabilpenari buruz baizik, eta ez dago zalantzarik jokoak izaten direla erabilpenaren aldetik erakargarrienak eta agian urtegarai honetarako aproposenak.

Ezaguna izango duzue gehienok URKATUA izenez ezagutzen den jokoak. Jolaskidetako batek (kasu honetan ordenadorea izango da) hitz bat aukeratuko du

eta besteak letrak botatzen joan behar-ko du; esaten duen letra bakoitzeko

asmatu duen ala ez eta asmatu duen kasuan letra hori zenbat aldiz eta hitzaren zein posiziotan azaltzen den zehatz-mehatz adierazi beharko zaio.

Jokoaren hasieran urka bat prestatuko da eta jokoan zehar egiten den huts bakoitzeko gorputzaren zati bat irudikatuko da, harik eta hitzaren letra guztiak asmatu edota gorputz osoa irudikatu ondoren urkatua geratzen den arte.

Joko bat izan arren hiztegia lantzeko oso erabilgarria gerta daitekeela uste dugu, baina eritzi hori zuen, erabiltzaileen, esku geratzen da.

Beraz, animo eta jolastu!

```

10 REM URKATUA
20 SCREEN 2:CLS:KEY OFF
30 DIM HITZA$(20,2),A(3520),B(160)
40 LOCATE 2,15:PRINT "HITZA ASMATU EDO ..."
50 LOCATE 3,15:PRINT "-----"
60 LINE (421,138)-(615,182),,B:PAINT (422,140),1,1
70 LINE (607,137)-(615,10),,B:PAINT (608,135),1,1
80 LINE (445,13)-(615,10),,B:PAINT (457,11),1,1
90 LINE (525,14)-(525,33):CIRCLE (525,44),10,,,3/2
100 RESTORE 600:RANDOMIZE TIMER
110 ALE=INT(RND*40)+1
120 FOR I=1 TO ALE:READ H$:NEXT I
130     FOR I=1 TO LEN(H$)
140         HITZA$(I,1)=MID$(H$,I,1):HITZA$(I,2)="0"
150     NEXT I
160 FOR I=1 TO LEN(H$):LOCATE 13,10+I*2:PRINT CHR$(223):NEXT I
170 GAIZKI=0:ASMAT$="F"
180 WHILE ASMAT$="F" AND GAIZKI<7
190     LOCATE 20,10:PRINT SPC(30):LOCATE 20,10:PRINT "Sartu letra bat: ";
200     TE$=INKEY$:WHILE TE$<>"":TE$=INKEY$:WEND
210     WHILE TE$="":TE$=INKEY$:WEND:PRINT TE$
220     IF TE$>"Z" THEN TE=ASC(TE$)-32:TE$=CHR$(TE)
230     IF TE$>"A" AND TE$<="Z" THEN LOCATE 25,10:PRINT SPC(30):GOTO 260
240     LOCATE 25,10:PRINT "Letra izan behar du!!!":BEEP:BEEP:BEEP

```

```

250      LOCATE 20,10:PRINT SPC(20)::GOTO 190
260 FOR I=1 TO LEN(H$)
270     IF HITZA$(I,1)<>TES$ THEN 290
280     LOCATE 12,10+I*2:PRINT HITZA$(I,1):HITZA$(I,2)="1":ASMAT$="E"
290 NEXT I
300 IF ASMAT$="E" THEN 390
310     GAIZKI=GAIZKI+1
320     IF GAIZKI=1 THEN LINE (514,55)-(534,66),,B:PAINT (516,56),1,1:GOTO 390
330     IF GAIZKI=2 THEN LINE (527,67)-(521,71),,B:PAINT (522,68),1,1:GOTO 390
340     IF GAIZKI=3 THEN LINE (501,72)-(547,104),,B:PAINT (502,73),1,1:GOTO 390
350     IF GAIZKI=4 THEN LINE (506,105)-(514,137),,B:PAINT (508,107),1,1:
        LINE (534,105)-(542,137),,B:PAINT (535,106),1,1:GOTO 390
360     IF GAIZKI=5 THEN LINE (490,72)-(501,76),,B:PAINT (492,74),1,1:
        LINE (490,72)-(496,102),,B:PAINT (492,100),1,1:
        LINE (558,72)-(548,76),,B:PAINT (550,74),1,1
370     IF GAIZKI=5 THEN LINE (558,72)-(552,102),,B:PAINT (553,100),1,1:GOTO 390
380     IF GAIZKI=6 THEN CIRCLE (525,44),10,0,,,3/2:LINE (525,33)-(525,55):
        CIRCLE (525,69),10,,,1/4
390     FOR I=1 TO LEN(H$)
400         IF HITZA$(I,2)="0" THEN ASMAT$="F"
410     NEXT I
420 WEND
430 LOCATE 20,10:PRINT SPC(30):LOCATE 19,10
440 IF ASMAT$="E" THEN PRINT "ONGI!!! Hitza asmatu duzu.":GOTO 520
450 GET (490,55)-(558,105),A:GET (490,77)-(496,102),B:
    PUT (490,55),A:PUT (490,75),A,PSET:CIRCLE (525,89),10,0,,,1/4
460 LINE (525,14)-(525,75):FOR I=1 TO 3:LINE (519,85+I)-(529,85+I):NEXT I:LINE (490,92)-(558,92)
470 FOR I=1 TO 6
480     IF I MOD 2<>0 THEN Y1=97:Y2=66 ELSE Y1=66:Y2=97
490     PUT (490,Y1),B:PUT (552,Y1),B:FOR J=1 TO 40:NEXT J:PUT (490,Y2),B:
        PUT (552,Y2),B:FOR J=1 TO 40:NEXT J
500 NEXT I
510 SOUND 300,3:PRINT "Ez duzu hitza asmatu!!!":LOCATE 21,10:PRINT "Hitza ";H$;" zen."
520 END
530 REM
600 DATA "ABIADURA","AZELERAZIO","ORDENADORE","ALKONDARA","GURASO"
610 DATA "LEHENGUSU","ARRAUTZA","ESKAILERA","TELEFONO","ASTEARTE","ASTEAZKEN"
620 DATA "ASTELEHEN","HILABETE","UDABERRI","PERTSONA","TXIMELETA","SUKALDE"
630 DATA "TELEBISTA","ARRATSALDE","BIZIMODU","EGURALDI","KILOMETRO","LAGUNTZA"
640 DATA "MAITASUN","ORDUBETE","AILEGATU","BAZKALDU","BEGIRATU","ERAKUTSI"
650 DATA "HARRAPATU","IRAKURRI","KONPONDU","BAKARRIK","BEHARBADA","BEREHALA"
660 DATA "BITARTEAN","ORAINTXE","ZERGIATIK","EUSKALDUN","AUTOMOBIL"

```

Programaren azalpena

10-90	Programaren sarrera; urka prestatzea.	300-380	siziotan agertzen den markatu taulan eta pantailan idatzi.
100-160	Datuetatik hitz bat aleatorioki hautatu; taulan karaktereka gorde eta letrak asmatu gabe bezala markatzen dira.	390-420	Huts egiteari dagokion irudizatia irudikatzea. 6 hutsegite gehienez.
170-420	Programaren ziklo nagusia. Ziklo hau 7 akats edo hutsegite egiten direnean edota hitza asmatzen denean amaituko da.	430-510	Jokoaren emaitza azaltzea.
190-250	Erabiltzaileak sakatu duen teklaren kontrola; letra dela egiaztatu eta maiuskuletara pasa.	440	Hitza asmatu den kasua.
260-290	Sartutako letra hautatutako hitzean agertzen bada, zein po-	450-510	Hitza asmatu ez den kasua.
		600-660	Programan erabiliko diren hitzak, datuak.